TwoPendula

Let A = K[t, L_1, L_2, m_1, m_2, M, g] be the commutative polynomial ring in t, L_1, L_2, m_1, m_2, M, g and let us define the Ore algebra A = A⟨∂⟩, where ∂ is differential operator defined by $\partial f(t) = f'(t)$.

```
A = OreAlgebra[t, Der[t], L1, L2, m1, m2, M, g]
```

$\mathbb{K}[t, L_1, L_2, m_1, m_2, M, g][D_t; 1, D_t]$

We consider the matrix $R \in A^{3\times 4}$ given by

```
mat = {{m1 L1 Der[t]^2, m2 L2 Der[t]^2, (M + m1 + m2) Der[t]^2, -1},
    {m1 L1^2 Der[t]^2 - m1 L1 g, 0, m1 L1 Der[t]^2, 0},
    {0, m2 L2^2 Der[t]^2 - m2 L2 g, m2 L2 Der[t]^2, 0}
  };
R = ToOrePolynomial[mat, A];
MatrixForm[R]
```

$$\begin{pmatrix} D_t^2 L_1 m_1 & D_t^2 L_2 m_2 & D_t^2 m_1 + D_t^2 m_2 + D_t^2 M & -1 \\ D_t^2 L_1^2 m_1 - L_1 m_1 g & 0 & D_t^2 L_1 m_1 & 0 \\ 0 & D_t^2 L_2^2 m_2 - L_2 m_2 g & D_t^2 L_2 m_2 & 0 \end{pmatrix}$$

and the left A-module, finitely presented by R, is M = $A^{1\times 4}/A^{1\times 3} R$.
Let us compute the adjoint of the system

```
MatrixForm[Radj = Involution[R, A]]
```

$$\begin{pmatrix} D_t^2 L_1 m_1 & D_t^2 L_1^2 m_1 - L_1 m_1 g & 0 \\ D_t^2 L_2 m_2 & 0 & D_t^2 L_2^2 m_2 - L_2 m_2 g \\ D_t^2 m_1 + D_t^2 m_2 + D_t^2 M & D_t^2 L_1 m_1 & D_t^2 L_2 m_2 \\ -1 & 0 & 0 \end{pmatrix}$$

Let us check whether or not the left A-module M is torsion-free, i.e. whether or not the system admits some autonomous elements.

```
{Ann, Rp, Q} = Exti[Radj, A, 1];
MatrixForm[Ann]
```

$$\begin{pmatrix} L_1 L_2 & 0 & 0 \\ 0 & L_2 m_2 & 0 \\ 0 & 0 & L_1 m_1 \end{pmatrix}$$

```
MatrixForm[Rp]
```

$$\begin{pmatrix} m_1 g & m_2 g & D_t^2 M & -1 \\ 0 & -D_t^2 L_2 + g & -D_t^2 & 0 \\ -D_t^2 L_1 + g & 0 & -D_t^2 & 0 \end{pmatrix}$$

```
Q
```

$\{\{D_t^4 L_2 - D_t^2 g\}, \{D_t^4 L_1 - D_t^2 g\}, \{-D_t^4 L_1 L_2 + D_t^2 L_1 g + D_t^2 L_2 g - g^2\},$
$\{-D_t^6 L_1 L_2 M + D_t^4 L_1 m_2 g + D_t^4 L_1 M g + D_t^4 L_2 m_1 g + D_t^4 L_2 M g - D_t^2 m_1 g^2 - D_t^2 m_2 g^2 - D_t^2 M g^2\}\}$

Since the entries of the first matrix are constants (zero order operators in D_t), if $L_1 \neq 0$, $L_2 \neq 0$, $m_1 \neq 0$, $m_2 \neq 0$ then the module is torsion free. Let us now compute the obstructions for M to be projective:

```
Obs = ObstructionToProjectiveness[R, A]
```

$$\{\{\{L_1 L_2 m_1 m_2\}\}, \{\{-L_1 g^2 + L_2 g^2\}, \{-D_t^2 L_2 g + g^2\}, \{-D_t^2 L_1 g + g^2\}, \{D_t^4 L_2 - D_t^2 g\}, \{D_t^4 L_1 - D_t^2 g\}\}\}$$

Let us deduce the obstructions in the parameters L_1, L_2, m_1, m_2, M, g for M to be a projective left A-module.

```
obs = OreIntersection[Flatten@Obs, {L1, L2, m1, m2, M, g}, A]
```

$$\{-L_1 g^2 + L_2 g^2, m_1 m_2 g^2, L_1 L_2 m_1 m_2\}$$

We obtain that M is a projective left A-module, i.e. the system is flat, if and only if $L_1 \neq 0$, $L_2 \neq 0$, $m_1 \neq 0$, $m_2 \neq 0$, $g \neq 0$ and $L_1 \neq L_2$.

If $L_1 L_2 m_1 m_2 g \neq 0$ and $L_1 \neq L_2$, let us compute flat output.

```
Factor[iOrePolynomialToNormal[obs]]
```

$$\{-g^2 (L_1 - L_2), g^2 m_1 m_2, L_1 L_2 m_1 m_2\}$$

```
B = OreAlgebra[t, Der[t]]
```

$\mathbb{K}[t][D_t; 1, D_t]$

```
T = LeftInverse[Q, B]
```

$$\left\{\left\{-\frac{L_1^2}{g^2 (L_1 - L_2)}, \frac{L_2^2}{g^2 (L_1 - L_2)}, -\frac{1}{g^2}, 0\right\}\right\}$$

A flat output ξ is then defined by:

```
ApplyMatrix[T, {x1[t], x2[t], x3[t], u[t]}][[1]]
```

$$\frac{-L_1^2 x_1[t] - L_1 x_3[t] + L_2 (L_2 x_2[t] + x_3[t])}{g^2 (L_1 - L_2)}$$

Finally a parametrization of the flat system is defined by

```
Thread[{x1[t], x2[t], x3[t], u[t]} -> ApplyMatrix[Q, {ξ[t]}]]
```

$$\{x_1[t] \to -g \xi''[t] + L_2 \xi^{(4)}[t], x_2[t] \to -g \xi''[t] + L_1 \xi^{(4)}[t],$$
$$x_3[t] \to -g^2 \xi[t] + g L_2 \xi''[t] + L_1 (g \xi''[t] - L_2 \xi^{(4)}[t]),$$
$$u[t] \to -g^2 (M + m_1 + m_2) \xi''[t] + g L_2 (M + m_1) \xi^{(4)}[t] + L_1 (g (M + m_2) \xi^{(4)}[t] - M L_2 \xi^{(6)}[t])\}$$

If $L_1 = L_2$, then let us compute the autonomous elements of the new system. To do that let us introduce a new Ore algebra B, defined by

```
B = OreAlgebra[t, Der[t], CoefficientNormal → (Expand[# /. L1 → L2] &)]
```

$\mathbb{K}[t][D_t; 1, D_t]$

the matrix S of differential operators given by

```
MatrixForm[S = ToOrePolynomial[mat, B]]
```

$$\begin{pmatrix} L_2 m_1 D_t^2 & L_2 m_2 D_t^2 & (M + m_1 + m_2) D_t^2 & -1 \\ L_2^2 m_1 D_t^2 - g L_2 m_1 & 0 & L_2 m_1 D_t^2 & 0 \\ 0 & L_2^2 m_2 D_t^2 - g L_2 m_2 & L_2 m_2 D_t^2 & 0 \end{pmatrix}$$

and the left B-module N finitely presented by S. Let us compute the torsion elements of N, i.e. the autnomous elements of the corresponding system.

```
MatrixForm[Sadj = Involution[S, B]]
```

$$\begin{pmatrix} L_2 m_1 D_t^2 & L_2^2 m_1 D_t^2 - g L_2 m_1 & 0 \\ L_2 m_2 D_t^2 & 0 & L_2^2 m_2 D_t^2 - g L_2 m_2 \\ (M + m_1 + m_2) D_t^2 & L_2 m_1 D_t^2 & L_2 m_2 D_t^2 \\ -1 & 0 & 0 \end{pmatrix}$$

```
{Ann, Sp, Q} = Exti[Sadj, B, 1];
MatrixForm[Ann]
```

$$\begin{pmatrix} L_2 D_t^2 - g & 0 & 0 \\ 0 & -L_2 D_t^2 + g & 0 \\ 0 & 0 & L_2 D_t^2 - g \end{pmatrix}$$

```
MatrixForm[Sp]
```

$$\begin{pmatrix} 1 & -1 & 0 & 0 \\ 0 & g m_1 + g m_2 & M D_t^2 & -1 \\ 0 & -M L_2 D_t^2 + (g M + g m_1 + g m_2) & 0 & -1 \end{pmatrix}$$

```
MatrixForm[Q]
```

$$\begin{pmatrix} -D_t^2 \\ -D_t^2 \\ L_2 D_t^2 - g \\ M L_2 D_t^4 + (-g M - g m_1 - g m_2) D_t^2 \end{pmatrix}$$

Since the first matrix is not identity, the system is not controllable (N is not torsion free) and we have autonomous elements, defined by the rows of Sp, namely θ, defined by

```
ApplyMatrix[Sp, {x1[t], x2[t], x3[t], u[t]}]
```

$\{x_1[t] - x_2[t],\ -u[t] + g\ (m_1 + m_2)\ x_2[t] + M\, x_3''[t],$
$\ -u[t] + g\ (M + m_1 + m_2)\ x_2[t] - M\, L_2\, x_2''[t]\}$

For instance, $\theta 1$ satisfies the equation

```
Thread[ApplyMatrix[{{Ann[[1, 1]]}}, {θ1[t]}] == 0]
```

$\{-g\, \theta 1[t] + L_2\, \theta 1''[t] == 0\}$

Finally, this result can be obtained directly by means of the command AutonomousElements

```
AutonomousElements[S, {x1[t], x2[t], x3[t], u[t]}, θ, B, Relations → True]
```

$$\Big\{\{\theta[1][t] \to x_1[t] - x_2[t],\ \theta[2][t] \to -u[t] + g\,(m_1 + m_2)\,x_2[t] + M\,x_3''[t],\ \theta[3][t] \to -u[t] + g\,(M + m_1 + m_2)\,x_2[t] - M\,L_2\,x_2''[t]\},\ \{-g\,\theta[1][t] + L_2\,\theta[1]''[t] == 0,\ g\,\theta[2][t] - L_2\,\theta[2]''[t] == 0,\ -g\,\theta[3][t] + L_2\,\theta[3]''[t] == 0\},$$

$$\Big\{\frac{(M + m_1 + m_2)\,\theta[2][t] - m_2\,\theta[3][t] + m_1\,(-\theta[3][t] + M\,L_2\,\theta[1]''[t])}{M} == 0,$$

$$\frac{L_2\,m_1\,(-g\,M\,\theta[1][t] + \theta[2][t] - \theta[3][t] + M\,L_2\,\theta[1]''[t])}{M} == 0,$$

$$\frac{L_2\,m_2\,(\theta[2][t] - \theta[3][t])}{M} == 0\Big\}\Big\}$$

The controllable part of the system is defined by:

```
Thread[ApplyMatrix[Sp, {x1[t], x2[t], x3[t], u[t]}] == 0]
```

$$\{x_1[t] - x_2[t] == 0,\ -u[t] + g\,(m_1 + m_2)\,x_2[t] + M\,x_3''[t] == 0,\ -u[t] + g\,(M + m_1 + m_2)\,x_2[t] - M\,L_2\,x_2''[t] == 0\}$$

Finally let us compute a flat output of the controllable part:

```
ApplyMatrix[LeftInverse[Q, B], {x1[t], x2[t], x3[t], u[t]}][[1]]
```

$$-\frac{L_2\,x_2[t] + x_3[t]}{g}$$